F07BSF (CGBTRS/ZGBTRS) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07BSF (CGBTRS/ZGBTRS) solves a complex band system of linear equations with multiple right-hand sides, $AX = B$, $A^TX = B$ or $A^HX = B$, where A has been factorized by F07BRF (CGBTRF/ZGBTRF).

2 Specification

```
 SUBROUTINE F07BSF(TRANS, N, KL, KU, NRHS, AB, LDAB, IPIV, B, LDB,
1                  INFO)
 ENTRY      cgbtrs(TRANS, N, KL, KU, NRHS, AB, LDAB, IPIV, B, LDB,
1                  INFO)
 INTEGER           N, KL, KU, NRHS, LDAB, IPIV(*), LDB, INFO
 complex           AB(LDAB,*), B(LDB,*)
 CHARACTER*1       TRANS
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To solve a complex band system of linear equations $AX = B$, $A^TX = B$ or $A^HX = B$, this routine must be preceded by a call to F07BRF (CGBTRF/ZGBTRF) which computes the LU factorization of A as $A = PLU$. The solution is computed by forward and backward substitution.

If TRANS = 'N', the solution is computed by solving $PLY = B$ and then $UX = Y$.

If TRANS = 'T', the solution is computed by solving $U^TY = B$ and then $L^TP^TX = Y$.

If TRANS = 'C', the solution is computed by solving $U^HY = B$ and then $L^HP^TX = Y$.

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

1: TRANS — CHARACTER*1 *Input*

On entry: indicates the form of the equations as follows:

if TRANS = 'N', then $AX = B$ is solved for X;
if TRANS = 'T', then $A^TX = B$ is solved for X;
if TRANS = 'C', then $A^HX = B$ is solved for X.

Constraint: TRANS = 'N', 'T' or 'C'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: KL — INTEGER *Input*

On entry: k_l, the number of sub-diagonals within the band of A.

Constraint: KL ≥ 0.

4: KU — INTEGER *Input*

On entry: k_u, the number of super-diagonals within the band of A.

Constraint: KU ≥ 0.

5: NRHS — INTEGER *Input*

On entry: r the number of right-hand sides.

Constraint: NRHS ≥ 0.

6: AB(LDAB,$*$) — ***complex*** array *Input*

Note: the second dimension of the array AB must be at least max(1,N).

On entry: the LU factorization of A, as returned by F07BRF (CGBTRF/ZGBTRF).

7: LDAB — INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BSF (CGBTRS/ZGBTRS) is called.

Constraint: LDAB $\geq 2 \times$ KLM + KU + 1.

8: IPIV($*$) — INTEGER array *Input*

Note: the dimension of the array IPIV must be at least max(1,N).

On entry: the pivot indices, as returned by F07BRF (CGBTRF/ZGBTRF).

9: B(LDB,$*$) — ***complex*** array *Input/Output*

Note: the second dimension of the array B must be at least max(1,NRHS).

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

10: LDB — INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07BSF (CGBTRS/ZGBTRS) is called.

Constraint: LDB $\geq$ max(1,N).

11: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$|E| \leq c(k)\epsilon|L||U|,$$

$c(k)$ is a modest linear function of $k = k_l + k_u + 1$, and ϵ is the ***machine precision***. This assumes $k \ll n$.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(k)\text{cond}(A, x)\epsilon$$

where $\mathrm{cond}(A,x) = \| \, |A^{-1}||A||x| \, \|_\infty / \|x\|_\infty \leq \mathrm{cond(A)} = \| \, |\mathrm{A}^{-1}||\mathrm{A}| \, \|_\infty \leq \kappa_\infty(\mathrm{A})$. Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$, and $\mathrm{cond}(A^H)$ (which is the same as $\mathrm{cond}(A^T)$) can be much larger (or smaller) than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07BVF (CGBRFS/ZGBRFS), and an estimate for $\kappa_\infty(A)$ can be obtained by calling F07BUF (CGBCON/ZGBCON) with NORM = 'I'.

8 Further Comments

The total number of real floating-point operations is approximately $8n(2k_l + k_u)r$, assuming $n \gg k_l$ and $n \gg k_u$.

This routine may be followed by a call to F07BVF (CGBRFS/ZGBRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07BEF (SGBTRS/DGBTRS).

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.65+2.26i & -2.05-0.85i & 0.97-2.84i & 0.00+0.00i \\ 0.00+6.30i & -1.48-1.75i & -3.99+4.01i & 0.59-0.48i \\ 0.00+0.00i & -0.77+2.83i & -1.06+1.94i & 3.33-1.04i \\ 0.00+0.00i & 0.00+0.00i & 4.48-1.09i & -0.46-1.72i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -1.06+21.50i & 12.85+\ 2.84i \\ -22.72-53.90i & -70.22+21.57i \\ 28.24-38.60i & -20.73-\ 1.23i \\ -34.56+16.73i & 26.01+31.97i \end{pmatrix}.$$

Here A is nonsymmetric and is treated as a band matrix, which must first be factorized by F07BRF (CGBTRF/ZGBTRF).

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07BSF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KLMAX, KUMAX, LDAB, NRHMAX, LDB
      PARAMETER        (NMAX=8,KLMAX=8,KUMAX=8,LDAB=2*KLMAX+KUMAX+1,
     +                 NRHMAX=NMAX,LDB=NMAX)
      CHARACTER        TRANS
      PARAMETER        (TRANS='N')
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, K, KL, KU, N, NRHS
*     .. Local Arrays ..
      complex          AB(LDAB,NMAX), B(LDB,NRHMAX)
      INTEGER          IPIV(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         cgbtrf, cgbtrs, X04DBF
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
```

```
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07BSF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS, KL, KU
      IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX .AND. KL.LE.KLMAX .AND. KU.LE.
     +    KUMAX) THEN
*
*        Read A and B from data file
*
         K = KL + KU + 1
         READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,N)
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Factorize A
*
         CALL cgbtrf(N,N,KL,KU,AB,LDAB,IPIV,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Compute solution
*
            CALL cgbtrs(TRANS,N,KL,KU,NRHS,AB,LDAB,IPIV,B,LDB,INFO)
*
*           Print solution
*
            IFAIL = 0
            CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
     +                  'Solution(s)','Integer',RLABS,'Integer',CLABS,
     +                  80,0,IFAIL)
         ELSE
            WRITE (NOUT,*) 'The factor U is singular'
         END IF
      END IF
      STOP
*
      END
```

9.2 Program Data

```
F07BSF Example Program Data
  4  2  1  2                                        :Values of N, NRHS, KL and KU
 (-1.65, 2.26) (-2.05,-0.85) ( 0.97,-2.84)
 ( 0.00, 6.30) (-1.48,-1.75) (-3.99, 4.01) ( 0.59,-0.48)
               (-0.77, 2.83) (-1.06, 1.94) ( 3.33,-1.04)
                             ( 4.48,-1.09) (-0.46,-1.72)  :End of matrix A
 ( -1.06, 21.50) ( 12.85,  2.84)
 (-22.72,-53.90) (-70.22, 21.57)
 ( 28.24,-38.60) (-20.73, -1.23)
 (-34.56, 16.73) ( 26.01, 31.97)                         :End of matrix B
```

9.3 Program Results

```
F07BSF Example Program Results

Solution(s)
                     1                  2
1  (-3.0000, 2.0000) ( 1.0000, 6.0000)
2  ( 1.0000,-7.0000) (-7.0000,-4.0000)
3  (-5.0000, 4.0000) ( 3.0000, 5.0000)
4  ( 6.0000,-8.0000) (-8.0000, 2.0000)
```